Interpolation over triangle and inverse mapping

Triangular coordinates

In[1]:=
```
Tmat = {{1, 1, 1}, {x1, x2, x3}, {y1, y2, y3}};
```

In[2]:=
```
StringForm["``=``.``", {1, x, y} // MatrixForm, Tmat // MatrixForm, {ζ1, ζ2, ζ3} // MatrixForm]
```

Out[2]=
$$\begin{pmatrix} 1 \\ x \\ y \end{pmatrix} = \begin{pmatrix} 1 & 1 & 1 \\ x1 & x2 & x3 \\ y1 & y2 & y3 \end{pmatrix} . \begin{pmatrix} \zeta 1 \\ \zeta 2 \\ \zeta 3 \end{pmatrix}$$

Area of triangle given vertices

In[3]:=
```
area2 = Det[Tmat] // Simplify (*two times the area*)
```

Out[3]=
$$x3\,(y1 - y2) + x1\,(y2 - y3) + x2\,(-y1 + y3)$$

Inverse of transformation matrix

In[4]:=
```
Tinv = Inverse[Tmat] /. {Det[Tmat] → A2} // MatrixForm
```

Out[4]//MatrixForm=
$$\begin{pmatrix} \frac{-x3\,y2+x2\,y3}{A2} & \frac{y2-y3}{A2} & \frac{-x2+x3}{A2} \\ \frac{x3\,y1-x1\,y3}{A2} & \frac{-y1+y3}{A2} & \frac{x1-x3}{A2} \\ \frac{-x2\,y1+x1\,y2}{A2} & \frac{y1-y2}{A2} & \frac{-x1+x2}{A2} \end{pmatrix}$$

Inverse mapping: computation of the triangle coordinates given (x,y)

In[5]:=
```
res = Inverse[Tmat].{1, x, y} /. {Det[Tmat] → A2} // Simplify
```

Out[5]=
$$\left\{ \frac{x3\,(y - y2) + x\,(y2 - y3) + x2\,(-y + y3)}{A2}, \right.$$
$$\left. \frac{x1\,y - x3\,y - x\,y1 + x3\,y1 + x\,y3 - x1\,y3}{A2}, \frac{x2\,(y - y1) + x\,(y1 - y2) + x1\,(-y + y2)}{A2} \right\}$$

In[6]:=
```
A2Pretty = x1 (y2 - y3) + x2 (y3 - y1) + x3 (y1 - y2);
ζ1Pretty = (x (y2 - y3) + x2 (y3 - y) + x3 (y - y2)) / A2;
ζ2Pretty = (x1 (y - y3) + x (y3 - y1) + x3 (y1 - y)) / A2;
ζ3Pretty = (x1 (y2 - y) + x2 (y - y1) + x (y1 - y2)) / A2;
```

In[10]:=
```
Det[Tmat] == A2Pretty // Simplify
```

Out[10]=
```
True
```

In[11]:=
```
res[[1]] == ζ1Pretty // Simplify
```

Out[11]=
```
True
```

In[12]:=
```
res[[2]] == ζ2Pretty // Simplify
```

Out[12]=
```
True
```

In[13]:=
```
res[[3]] == ζ3Pretty // Simplify
```

Out[13]=
```
True
```

Interpolation

Definition

In[14]:=
```
TriInterp[x_, y_, coords_, temperatures_] :=
  Module[{x1, y1, x2, y2, x3, y3, A2, ζ1, ζ2, ζ3, T1, T2, T3},
   {{x1, y1}, {x2, y2}, {x3, y3}} = Take[coords];
   {T1, T2, T3} = Take[temperatures];
   A2 = x1 (y2 - y3) + x2 (y3 - y1) + x3 (y1 - y2);
   ζ1 = (x (y2 - y3) + x2 (y3 - y) + x3 (y - y2)) / A2;
   ζ2 = (x1 (y - y3) + x (y3 - y1) + x3 (y1 - y)) / A2;
   ζ3 = (x1 (y2 - y) + x2 (y - y1) + x (y1 - y2)) / A2;
   Return[{ζ1, ζ2, ζ3}.{T1, T2, T3}]
  ];
```

In[15]:=
```
ClearAll[x, y, x1, y1, x2, y2, x3, y3, T1, T2, T3, coords, temperatures];
coords = {{x1, y1}, {x2, y2}, {x3, y3}};
temperatures = {T1, T2, T3};
TriInterp[x, y, coords, temperatures] // Simplify
TriInterp[x1, y1, coords, temperatures] // Simplify
TriInterp[x2, y2, coords, temperatures] // Simplify
TriInterp[x3, y3, coords, temperatures] // Simplify
```

Out[18]=
```
(T3 (-x1 y + x2 y + x y1 - x2 y1 - x y2 + x1 y2) + T2 (-x3 y - x y1 + x3 y1 + x1 (y - y3) + x y3) +
    T1 (-x2 y + x3 y + x y2 - x3 y2 - x y3 + x2 y3)) / (x3 (y1 - y2) + x1 (y2 - y3) + x2 (-y1 + y3))
```

Out[19]=
```
T1
```

Out[20]=
```
T2
```

Out[21]=
```
T3
```

Plot

In[42]:=
```
coords = {{0, 0}, {1.5, 0}, {0, 1}};
temps = {3, 2, 1};
tri = Triangle[coords];
f3d = Plot3D[TriInterp[x, y, coords, temps], {x, y} ∈ tri];
t3d = Graphics3D[{Yellow, Triangle[{##, Min[temps]} & @@@ tri[[1]]]}];
mpl = Manipulate[Show[f3d, t3d,
    Graphics3D[{Green, PointSize[0.03],
      Point[{xp + α, yp + α, TriInterp[xp + α, yp + α, coords, temps]}]}]
  ], {xp, 0, 1.5}, {yp, 0, 1}, {α, 0, 0.5}]
```

Out[47]=



In[48]:=
```
Export["/home/dorival/Downloads/manipulate.avi", mpl]
```

Out[48]= /home/dorival/Downloads/manipulate.avi